

Photocell and Calibration Method

<http://arduino.cc/en/Tutorial/Calibration>

<http://learn.adafruit.com/photocells>

This example demonstrates one technique for calibrating sensor input. The Arduino takes sensor readings for five seconds during the startup, and tracks the highest and lowest values it gets. These sensor readings during the first five seconds of the sketch execution define the minimum and maximum of expected values for the readings taken during the loop.

HARDWARE REQUIRED


Arduino board
1 LED
1 Photocell (LDR) or any analog sensor
1 10K ohm resistor
1 220 ohm resistor
breadboard

CIRCUIT


Analog sensor (e.g. potentiometer, light sensor) on analog input 2.
LED on digital pin 9.

Connect an LED to digital pin 9 with a 220 ohm current limiting resistor. Connect a photocell to 5V and then to analog pin 0 with a 10K ohm resistor as a reference to ground.

SCHEMATIC



IMAGE



CODE

```
// These constants won't change:
const int sensorPin = A0; // pin that the sensor is attached to
const int ledPin = 9; // pin that the LED is attached to

// variables:
int sensorValue = 0; // the sensor value
int sensorMin = 1023; // minimum sensor value
int sensorMax = 0; // maximum sensor value

void setup() {
  Serial.begin(9600);
  pinMode(ledPin, OUTPUT);
  // turn on LED to signal the start of the calibration period:
  pinMode(13, OUTPUT); // ONBOARD LED pin 13
  digitalWrite(13, HIGH);

  // calibrate during the first five seconds
  while (millis() < 5000) {
    sensorValue = analogRead(sensorPin);

    // record the maximum sensor value
    if (sensorValue > sensorMax) {
      sensorMax = sensorValue;
    }

    // record the minimum sensor value
    if (sensorValue < sensorMin) {
      sensorMin = sensorValue;
    }
  }

  // signal the end of the calibration period
  digitalWrite(13, LOW);
}

void loop() {
  // read the sensor:
  sensorValue = analogRead(sensorPin);
  Serial.print(sensorValue);
  Serial.print("\n");

  // apply the calibration to the sensor reading
  sensorValue = map(sensorValue, sensorMin, sensorMax, 0, 255);

  // in case the sensor value is outside the range seen during calibration
  sensorValue = constrain(sensorValue, 0, 255);
  Serial.print(sensorValue);

  // fade the LED using the calibrated value:
  analogWrite(ledPin, sensorValue);
}
```

/*

Calibration

Demonstrates one technique for calibrating sensor input. The sensor readings during the first five seconds of the sketch execution define the minimum and maximum of expected values attached to the sensor pin.

The sensor minimum and maximum initial values may seem backwards.

Initially, you set the minimum high and listen for anything

lower, saving it as the new minimum. Likewise, you set the maximum low and listen for anything higher as the new maximum.

The circuit:

* Analog sensor (photo cell will do) attached to analog input 0

* LED attached from digital pin 9 to ground

created 29 Oct 2008

By David A Mellis

modified 30 Aug 2011

By Tom Igoe

<http://arduino.cc/en/Tutorial/Calibration>

This example code is in the public domain.

*/